
Subject: Re: A distracting puzzle

Posted by [Stein Vidar Hagfors H\[1\]](#) on Tue, 25 Sep 2001 16:11:10 GMT

[View Forum Message](#) <> [Reply to Message](#)

If what's being sought here is only to distinguish which pixels have *some* area inside the polygon and which do not, wouldn't it be sufficient to check the corners? I.e., in a continuum of pixel coordinates, given corners with coordinates [0,0], [1,0], [1,1], [0,1], it can be checked whether each of those are inside versus outside any defined polygon. If one or more of the corners is inside, then some area is also inside..

I have included some simple-minded routines I wrote some years ago to check whether a point is inside or outside a polygon...

Stein Vidar

```
;; $Id: vectorangle.pro,v 1.1 1999/06/02 16:24:14 steinhhh Exp $
;; The angle between vector A & B
;; The angle that vector A needs to be rotated (counterclockwise) in order
;; to be parallell to B
```

```
FUNCTION vectorangle,x1,y1,x2,y2,zerovalue=zerovalue
```

```
    default,zerovalue,0.0
```

```
    dp = x1*x2 + y1*y2
    cp = x1*y2 - x2*y1
```

```
    ix = where(dp EQ 0 AND cp EQ 0)
    IF ix(0) EQ -1L THEN return,atan(cp,dp)*!radeg
```

```
    dp(ix) = 1.0
    res = atan(cp,dp)*!radeg
    res(ix) = zerovalue
    return,res
```

```
END
```

```
;; $Id: insidepolygon.pro,v 1.2 1999/06/02 16:25:59 steinhhh Exp $
;; Return true if the given point is inside the
;;
;;
```

```
;; Poly == [2,N]
```

```

FUNCTION insidepolygon,ip,x,y,$
    edge_is_inside=edge_is_inside

    IF size(ip,/type) NE 4 AND size(ip,/type) NE 5 THEN p = float(ip) $
    ELSE BEGIN
        copyback = 1
        p = temporary(ip)
    END

    np = (size(p))(2)

    x1 = p(0,*)-x
    y1 = p(1,*)-y
    x2 = shift(p(0,*),0,-1)-x
    y2 = shift(p(1,*),0,-1)-y

    zeroval = 1e5
    theta = vectorangle(x1,y1,x2,y2,zerovalue=zeroval)

    ix = where(theta EQ zeroval $
        OR abs(theta-180.0d) LT 1e-4 $
        OR abs(theta+180.0d) LT 1e-4,count)
    IF count GT 0 THEN BEGIN
        result = keyword_set(edge_is_inside)
        GOTO,finished
    END

    ;; Test for those....

    result = abs(total(theta)) GT 180.0

finished:

    IF copyback THEN ip = temporary(p)

    return,result
END

```
